

Problem 33

The fraction $49/98$ is a curious fraction, as an inexperienced mathematician in attempting to simplify it may incorrectly believe that $49/98 = 4/8$, which is correct, is obtained by cancelling the 9s.

We shall consider fractions like, $30/50 = 3/5$, to be trivial examples.

There are exactly four non-trivial examples of this type of fraction, less than one in value, and containing two digits in the numerator and denominator.

If the product of these four fractions is given in its lowest common terms, find the value of the denominator.

Solution

```
In[31]:= twoDigs = Complement[Range[10, 99], Range[10, 100, 10]];

In[26]:= whatWeSeek[a_, b_] := With[{digsA = IntegerDigits[a], digsB = IntegerDigits[b]},
  With[{shared = Intersection[digsA, digsB]},
    If[shared === {}, False,
      (*otherwise shared contains one element, since a≠b*)
      If[Count[digsA, shared[[1]]] == 1,
        If[Count[digsB, shared[[1]]] == 1,
          FromDigits@Complement[digsA, shared] /
            FromDigits@Complement[digsB, shared] == a / b,

          FromDigits@Complement[digsA, shared] / shared[[1]] == a / b
        ],
      (*at this point, a is just a digit repeated twice*)
      If[Count[digsB, shared[[1]]] == 1,
        shared[[1]] / FromDigits@Complement[digsB, shared] == a / b,
        (*at this point, both a and b are the same digit repeated twice*)
        False]]
  ]]]

In[43]:= allowed = Select[Sort /@ Subsets[twoDigs, {2}], whatWeSeek @@ # &]

Out[43]:= {{16, 64}, {19, 95}, {26, 65}, {49, 98}}

In[47]:= Denominator[Divide @@ (Times @@@ Transpose[allowed])]

Out[47]:= 100
```